An active air spring suspension system without shock absorber on a quarter vehicle model

by

Mitchell Alonzo Duncklee

A Thesis Submitted to the

Graduate Faculty in Partial Fulfillment of the

Requirements for the Degree of

MASTER OF SCIENCE

Department: Aerospace Engineering and Engineering Mechanics
Major: Engineering Mechanics

Iowa State University
Ames, Iowa

1995

Dedicated to my father

TABLE OF CONTENTS

LIST OF FIGURES

LIST OF SYMBOLS

β	Ratio of orifice diameter to upstream diameter
γ	Ratio of specific heats
Δ	A differential value from an initial condition
ρ	Density in slugs/cubic foot
ω	Natural frequency in rad/sec
ξ	Damping ratio
A	Area in feet squared
A_s	Cross-sectional area of a cylinder in feet squared
C	Discharge coefficient
D	Diameter in feet
K	Flow coefficient
P	Pressure in pounds/cubic foot
V	Volume of air spring in feet cubed
Y	Expansion factor
c	Damping coefficient in pound second/foot
g	Acceleration of gravity in feet/second squared
h	Height of the air spring in feet
k	Stiffness of a mechanical spring in pounds/foot
m	Mass in slugs
r	Pressure ratio
x	Displacement in feet
z	Terrain displacement in feet

SUMMARY

The material contained in this document deals with the use of a pneumatic spring in an active suspension system. The control function of the suspension was derived to minimize transmissibility between the terrain and the chassis. A variable area orifice for air flow control between the air spring and an accumulator provides the mechanism for active control.

A bench model of the system and a computational simulation model were developed to investigate system dynamics. The computer model was primarily used to study control laws for the active system. Simulation accuracy was obtained by adjusting various system parameters to force a match between theoretical and experimental behavior. Tuning exercises were restricted to the uncontrolled system studies. The matching of the two models required significant investigation of the mathematical modeling of system dynamics and the flow of air between the spring and accumulator.

A control law was synthesized that produced a significant reduction in transmissibility of the uncontrolled model over a wide range of driving frequencies.

INTRODUCTION

Most current vehicles use a mechanical spring plus damper to isolate the chassis from terrain perturbations. The traditional spring/shock absorber system has the inherent limitation of a design compromise involving ride, handling qualities, and static deflection. Reasonable suspended mass deflections during transient motions and wide load ranges require springs with high spring rates. On the other hand, reasonable ride qualities require springs with low spring rates. Standard shock absorbers used to control damping characteristics further degrade ride qualities over a broad range of vehicle operation. Thus, traditional suspension systems optimized for ride qualities provide marginal control performance and vice-versa.

The active suspension system concept of recent years appears to be the natural replacement for traditional suspension systems in cases where both ride and control qualities are critical. Active systems generally allow uncoupling of static deflection, stiffness, and damping parameters in the suspension system. Hence, an active system can provide states of high controllability or rideability as required by the instantaneous dynamic situation [1].

The ride and control characteristics of a vehicle are associated with the transmission of forces from the terrain to the suspended mass. Discomfort and fatigue are a result of high transmissibility [2]. Minimizing transmissibility to provide good ride qualities for all inputs is a goal for almost all active suspension systems. On the other hand, stiffness of the suspension system to provide good control at critical times requires high transmissibility. Thus, active systems must ultimately provide transmissibility levels consistent with instantaneous needs.

Undesired transmissibility comes from two dynamic conditions; a system near resonance with insufficient damping, or a system away from resonance with excessive damping [3]. Hence, there must always be compromises in the amount of damping in a conventional suspension. A suspension that contains variable system parameters provides a mechanism to address the damping problem. That is, from the ride quality perspective, a system near resonance could be modified so that it has a new resonance point. The system would then be operating away from resonance and the transmissibility would be lowered. Such a system would have no need for permanent damping, thus alleviating the need for a shock absorber.

The possibilities for changing dynamic system parameters are limited for suspensions based on mechanical springs because of fixed spring rates. The air spring is much more accommodating with respect to changing stiffness and damping properties, and thus, represents a good choice for use in an active suspension system.

Air springs have been around for a long time but have only recently gained favor for actual road vehicles. In an air spring system air is trapped within a rubber boot that is free to stretch or compress. Boot volume oscillates about its design value as the spring articulates. The change in volume causes a change in air pressure and the resulting spring force. In particular, if the boot is compressed, the volume decreases, which increases the air pressure and the corresponding force on the suspended mass. This is analogous to the increased force produced by a mechanical spring as it is compressed.

The stiffness of the air spring is determined by its volume. When supporting the same mass, an air spring with a large volume is less stiff than one with a small volume. Thus, if the volume can be controlled, it follows that the stiffness characteristics can be controlled.

Normally air springs have a fixed design volume. Volume changes could be realized, however, through the use of a controlled access accumulator. An accumulator is simply a volume of air to which the spring is connected. Thus, and accumulated spring would be less stiff than a non-accumulated spring. Variable stiffness is obtained by controlling the air flow rate between the spring and accumulator.

The utility of an accumulated air spring as a major component in an active suspension system can be investigated through quarter vehicle modeling; a technique commonly used to determine the properties of a suspension system. The term 'quarter vehicle' comes from taking a four-wheeled vehicle and dividing it into fourths where the model has one wheel and one suspension component. The performance of the entire vehicle is a logical extension of the quarter vehicle modeling.

HARDWARE MODEL

The quarter vehicle model used in this research effort was derived from a modification of a bench test configuration constructed by Dr. Jerry Vogel and Dr. Lennox Wilson, of Iowa State University, in earlier research. The configuration, as shown in Figure 1, contains standard components of a quarter vehicle model. Major components include a suspended mass, a suspension spring, a tire/axle assembly and a terrain driver.



Figure 1. Schematic of quarter vehicle

The air spring is a prototype from Firestone Company. It was designed to carry weights in the 150 to 300 pound range. It consists of a cylinder of rubber folded under and attached to a bell shaped piston. Figure 2 shows the simple structure of the air spring.

The system also contains an accumulator and a controller for regulating air flow between the spring and accumulator. The controller consists of a fast actuating air cylinder/piston combination placed in a duct between the accumulator and air spring. When the piston is open the spring is connected to the accumulator; when closed it is isolated from the accumulator. The cylinder/piston can be actuated either manually or by computer control.



Figure 2. Schematic of the air spring



Figure 3. System geometry

The terrain input is achieved by means of an offset circular cam that provides a near sinusoidal displacement of predetermined amplitude and driving frequency. The quarter vehicle model tire rides on the cam and is free to move unrestricted in the vertical direction. The cam is driven by a high torque hydraulic motor with continuously variable output. The hydraulic pump is driven by a constant speed, three-phase electric motor.

The wheel mass includes the tire, a fork, and an assembly that provides attachment to the air spring. The tire is an ordinary motorcycle wheel. The suspended mass includes the accumulator, the controller, and a saddle to hold various weight combinations.

The system masses are restricted to motion in the vertical direction. Linear ball bushings mounted in pillow blocks on the mass assemblies travel on twin hardened steel shafts to provide low-friction motion paths.

The displacement of the suspended mass from a fixed reference frame is measured by a displacement potentiometer. A second displacement potentiometer is used to measure the position of the suspended mass relative to the axle assembly. Displacement transducer outputs were obtained through a data acquisition board placed in a PC computer. The board takes input signals from the potentiometers, and provides output signals for the controller. The data access rate of the board was sufficiently fast so that no appreciable lag between the system and the computer output was observed.

The system also incorporates a height control valve that keeps the air spring at a preset design height for all suspended mass values. The valve contains a six second delay to prevent it from impacting system dynamics.

COMPUTER MODEL

A quarter vehicle simulation code was developed to provide better understanding of the pneumatic isolation system under consideration. The computational procedure incorporated a Runge-Kutta/predictor-corrector scheme for the numerical integration of the system differential equations of motion. The simulation algorithm structure also included provisions for control elements in minimizing transmissibility.

Preliminary partial system modeling and coding were undertaken on the isolated air spring/suspended mass components in an attempt to get a better understanding of the air spring dynamics. Tire dynamics and the impact of accumulation were not included in this portion of the investigation. The simplified system is shown in Figure 4.

The equation of motion for this system is as follows:

$$\ddot{x} - \frac{PA}{m} + g = 0 \tag{1}$$



Figure 4. Dynamic model for an air spring/suspended mass

If P is taken as the equilibrium spring gauge pressure plus a perturbation differential pressure, ΔP, the equation becomes:

$$P = P_i + \Delta P \tag{2}$$

where:

$$P_i A = mg \tag{3}$$

therefore,

$$\ddot{x} - \frac{\Delta P A}{m} = 0 \tag{4}$$

Assuming an isentropic process for the air within the spring, an expression for ΔP can be found:

$$P + P_\infty = \rho^\gamma \left(\frac{P_\infty}{\rho_\infty^\gamma} \right) \tag{5}$$

where,

$$\rho = \frac{m_{air}}{(V_i + \Delta V)} \tag{6}$$

combining 5 and 6 yields,

$$\Delta P = (P_i + P_\infty)\left[\left(1 + \frac{\Delta V}{V_i} \right)^{-\gamma} - 1 \right] \tag{7}$$

which can be simplified and substituted into Equation 4:

$$\Delta P = P_i\left(1 + \frac{P_\infty}{P_i} \right)\left[\left(1 + \frac{\Delta V}{V_i} \right)^{-\gamma} - 1 \right] \tag{8}$$

such that:

$$\ddot{x}+\frac{A}{m}(P_i)\left(1+\frac{P_\infty}{P_i}\right)\left[\left(1+\frac{\Delta V}{V_i}\right)^{-\gamma}-1\right]=0 \tag{9}$$

Further simplification gives:

$$\ddot{x}+g\left(1+\frac{P_\infty}{P_i}\right)\left[\left(1+\frac{\Delta V}{V_i}\right)^{-\gamma}-1\right]=0 \tag{10}$$

In the case of a constant cross-sectional area spring, it is possible to write Equation 10 in terms of x as follows.

$$\Delta V = Ax \tag{11}$$

$$\ddot{x}-g\left(1+\frac{P_\infty}{P_i}\right)\left[\left(1+\frac{Ax}{V_i}\right)^{-\gamma}-1\right]=0 \tag{12}$$

For Ax/V_i much less than unity, Equation 12 can be linearized using a binomial expansion:

$$\left(1+\frac{Ax}{V_i}\right)^{-\gamma}\approx 1+(-\gamma)\left(\frac{Ax}{V_i}\right)+\text{H.O.T.} \tag{13}$$

giving:

$$\ddot{x}+\left(1+\frac{P_\infty}{P_i}\right)\left(\frac{\gamma g A}{V_i}\right)x=0 \tag{14}$$

This can be compared to a conventional mechanical spring:

$$\ddot{x}+\frac{k}{m}x=0 \tag{15}$$

where

$$\omega_n^2 = \frac{k}{m} \tag{16}$$

Thus the natural frequency of the linearized and simplified air spring is:

$$\omega_n = \sqrt{\left(1+\frac{P_\infty}{P_i}\right)\left(\frac{gA_s\gamma}{V_i}\right)} \tag{17}$$

Where:

γ	Ratio of specific heats	V_i	Initial air spring volume
ρ	Instantaneous air density	ΔV	Volume perturbation from V_i
ρ_∞	Atmospheric air density	g	Acceleration of gravity
ω_n	Natural frequency	k	Mechanical spring stiffness
A	Cross-sectional area of spring	m	Suspended mass
P	Instantaneous gauge pressure	m_{air}	Mass of air in spring
P_i	Initial gauge pressure	x	Displacement of suspended mass
P_∞	Atmospheric pressure	$\ddot{x}$	Acceleration of suspended mass
ΔP	Pressure perturbation from P_i	H.O.T.	Higher ordered terms

If the suspended mass is relatively large, the spring equilibrium gauge pressure will be significantly higher than atmospheric pressure. Hence, for large masses and constant specific heat ratio, the linear system's natural frequency is primarily dependent on spring cross-sectional area, A, and the initial volume, V_i.

Simulating the full quarter vehicle system is difficult because of problems caused by complexities associated with the functionality of system components. For example, addition of an accumulator is not modeled by merely changing the design volume of the air spring. Air flows between the air spring and accumulator and must pass through an orifice of arbitrary size. Therefore, the impact of accumulation on system dynamics is a function of orifice size and can be influenced by plumbing geometry if air flow ducts are too restrictive. Hence, accurate system simulation must include modeling of the orifice flow as air moves between air spring and accumulator.

System damping represents another problem. Tests indicate that there is significant damping in the spring system even though no shock absorber is included in the apparatus. Friction and air bag side wall forces are the most likely cause. A viscous damping term was included in the system modeling for both the restricted and unrestricted accumulator flow cases. The damping constant was used as a tuning parameter for the simulation model.

Tire dynamics were also included in the quarter vehicle modeling. The tire was modeled as a standard linear second order dynamic system with damping and natural frequency characteristics adjusted to match the test apparatus values. The full quarter vehicle model with tire included is depicted in Figure 5.

Total system dynamics are represented by two second order differential equations; one for each mass. The associated free body diagrams and corresponding equations of motion are depicted in Figures 6 and 7.



Figure 5. Dynamic model of the quarter vehicle



Figure 6. Forces on the suspended mass



Figure 7. Forces on the tire mass

The equation of motion for the suspended mass is as follows:

$$\ddot{x}_s = \frac{PA}{m_s} - \frac{c_s}{m_s}\left(\dot{x}_s - \dot{x}_t\right) - g \tag{18}$$

The equation of motion for the tire mass is as follows:

$$\ddot{x}_t = \frac{c_s}{m_t}\left(\dot{x}_s - \dot{x}_t\right) - \frac{c_t}{m_t}\left(\dot{x}_s - \dot{x}_t\right) - \frac{k_t}{m_t}\left(x_t - z\right) - \frac{PA}{m_t} + \frac{m_s}{m_t}g \tag{19}$$

Where:

A	Spring piston area	$\ddot{x}_s$	Suspended mass acceleration
P	Instantaneous gauge pressure	$\dot{x}_s$	Suspended mass velocity
c_s	Air Spring damping coefficient	$\ddot{x}_t$	Tire mass acceleration
c_t	Tire damping coefficient	$\dot{x}_t$	Tire mass velocity

g	Acceleration of gravity	x_t	Tire mass displacement
k_t	Tire stiffness	$\dot{z}$	Terrain velocity
m_s	Suspended mass	z	Terrain displacement
m_t	Tire mass		

Equations 18 and 19 contain terms involving gauge pressure, P, which can be evaluated using Equations 5 and 6. The instantaneous air spring volume, a rather complex shape, is defined by means of a linear relationship derived through geometric solid modeling techniques. The mass of air contained in the air spring is evaluated by numerically integrating the mass flow equations defined by [4]:

$$\dot{m} = KYA\sqrt{\rho_1 \Delta P} \tag{20}$$

With:

$$K = \frac{C}{\sqrt{1-\beta^4}} \tag{22}$$

$$C = .6 \tag{22}$$

$$\beta = \frac{D_{orifice}}{D_1} \tag{23}$$

$$Y^2 = \left(\frac{\gamma r^{\frac{2}{\gamma}}}{\gamma - 1}\right)\left(\frac{1-\beta^4}{1-\beta^4 r^{\frac{2}{\gamma}}}\right)\left(\frac{1-r^{\frac{\gamma-1}{\gamma}}}{1-r}\right) \tag{24}$$

$$r = 1 - \frac{\Delta P}{P_1} \tag{25}$$

Where:

β	Ratio of orifice diameter to upstream diameter
γ	Ratio of specific heats
ρ_1	Upstream density

A	Orifice cross-sectional area
C	Discharge coefficient for a square edged orifice [5]
D_1	Upstream pipe diameter
$D_{orifice}$	Orifice diameter
K	Flow coefficient
P_1	Upstream pressure
ΔP	Difference between up and downstream pressures
Y	Expansion factor, (ideal compressible flow / ideal incompressible flow)
$\dot{m}$	Air flow between accumulator and air spring
r	Pressure ratio (downstream / upstream)

These equations are an empirical representation for the behavior of real flows. The discharge coefficient is used to account for losses in the system. Equations not involving the discharge coefficient are based on an isentropic, one-dimensional flow.

A complete listing of the computer code used in the quarter vehicle model is included in the appendix. The algorithm used in the simulation procedure is defined in the code documentation. An input file used to initiate simulation execution is included in the appendix following the code listing. The file contains values for various coefficients, constants, and initial values for system dynamic variables.

TESTING MODELS

The major function of the simulation code is to study control laws that lead to optimal isolation performance of the pneumatic system. Therefore the simulation system must replicate the dynamic performance of the actual hardware system. This was accomplished by adjusting modeling constants or coefficient values associated with the various physical components contained in the system. Some values are measurable or are known for the given operating conditions: specific heat ratio, orifice size, spring design height, tire and suspended mass magnitudes, etc. Others, such as tire damping and natural frequency, were more indistinct. Approximate values for these variables were obtained from simple tests and observations. Orifice discharge coefficient values were estimated from empirical databases presented in [5].

The air spring volume parameter was especially difficult to evaluate due to the inherent complex internal shape associated with the bag/piston combination. Estimates for volume magnitude versus spring height were obtained through solid modeling procedures using the CAD program, ProEngineer. A linear volume/spring height relationship was derived and incorporated in system modeling.

The variables available for tuning the simulation model are given in Table 1. A brief description of the effects of each parameter on system response are also included. Only the marked variables were used in tuning the model.

Table 1. Tuning variables and sensitivities

Tire natural frequency		Little effect if value greater than 50 rad/sec
Tire damping ratio		Almost no effect
Air spring damping coefficient	•	Flattens the transmissibility peak
Air spring volume function	•	Lager effect on natural frequency
Air spring piston diameter		Very small effect on natural frequency
Upstream piping diameter		Almost no effect
Air spring design height		Effects based on spring volume only
Accumulator volume	•	Effects open orifice case natural frequency

A variety of tests were undertaken to tune the simulation model to the actual hardware. Both the fully accumulated and the non-accumulated systems were included in the tests since these conditions represent the extremes in dynamic response.

Initial tests incorporated an impulse applied to the suspended mass to initiate an oscillation. Damped natural frequency and damping coefficient values were extracted from the response database. Test results indicated that the simulation output does not match the hardware response for the fully accumulated system. The damped natural frequency of the hardware system was considerably higher than that of the simulation system. The accumulator volume, a measurable value, had to be reduced significantly to increase the stiffness of the computer model.

The next set of tests included a sinusoidal displacement applied to the tire at various driving frequencies. System displacement transmissibility was evaluated and plotted for each frequency. The resulting transmissibility plots depict the dynamic characteristics of the system over a wide range of frequencies and allows a greater baseline for which simulation matching could occur.

The displacement transmissibility of the suspension system is defined as the ratio of suspended mass amplitude to tire mass amplitude. Transmissibility values were derived from test measurements of absolute displacement of the suspended mass and displacement of the suspended mass relative to the tire mass. The tire mass absolute displacement could be extracted from the two signals. Figures 8 gives an example of the output of the two displacement potentiometers. Figure 9 shows the absolute tire mass displacements after the signals have been modified.

For the purely sinusoidal motion the transmissibility can be determined by extracting the maximum and minimum values of the displacements. In the sample given, the transmissibility can be seen by superimposing the absolute displacements plots, as shown in Figure 10. Noting the suspended mass amplitude at 600 and the tire mass amplitude of 200 one would expect to calculate the transmissibility at about three. The actual signal amplitudes showed a transmissibility of 2.89.

Many driving frequencies were used to generate transmissibility curves for the hardware model for both the accumulated and non-accumulated systems. The simulation model was subjected to similar inputs. The transmissibility curves for the experiment and computer simulation were superimposed for direct comparisons. The combined plots provide an excellent mechanism for comparing simulation and hardware results. Final tuning for the simulation was accomplished by matching these plots.

The sample output in Figures 8 and 9 depict hardware results at a driving frequency of 1.3 Hz. The tuned computer model was executed at the same driving frequency to provide the computer model equivalent of the experimental results. Figures 11 and 12 show the simulation displacement and transmissibility curves. A detailed presentation of the comparative transmissibility plots is depicted in Figure 13.

The results shown in Figure 13 indicate that the simulation model was not capable of matching the high transmissibility of the non-accumulated system. Nor was it possible to replicate the low frequency response for the accumulated system. These inherent modeling problems were overcome by changing the mathematical modeling for the accumulator/spring air flow. The simulation is improved by adding a model component that restricts mass flow for pressure ratios below a threshold value. The threshold value was treated as a variable to be adjusted for better fitting the models. This model would allow a better match to both the accumulated and non-accumulated cases.

Figures 14 and 15 show the modified simulation displacements for the sample case. Figure 16 displays the transmissibility plots that show the improvements made by using the modified simulation which was used in the final model.



Figure 8. Example hardware system output



Figure 9. Example hardware absolute displacements



Figure 10. Example hardware transmissibility



Figure 11. Example simulation displacements



Figure 12. Example simulation transmissibility



Figure 13. Transmissibility plot for unmodified simulation



Figure 14. Modified simulation displacements



Figure 15. Modified simulation transmissibility



Figure 16. Transmissibility plot for modified simulation

CONTROLLING MODELS

A primary goal for the active suspension in this research is to minimize transmissibility for improved rideability. It is understood that this is not a system to provide good handling characteristics for a vehicle. Thus, transmissibility minimization is only a part of an overall practical active suspension system.

Two control laws were tested on the hardware and computer quarter vehicle models. The laws were very simple to implement because only the suspended mass absolute displacement was used as a control input.

The first control law was used in previous research on the vehicle model done by Dr. Vogel and Dr. Wilson. A control law was derived from an investigation of a mechanical system without damping and a time varying spring rate. The spring rate was manipulated in such a way as to match the system to a system with critical damping. The final law was an approximation to the calculated variable spring rates. The control law requires an open or closed accumulator orifice based on the position of the suspended mass. If the mass is located above the equilibrium position, or positive, the orifice was closed. If the mass was below equilibrium the orifice was open. Figure 17 demonstrates the control law for the simulation.

Figure 18 shows the transmissibility plots for the control law. A notable point to the control law is its behavior at high driving frequencies. A second resonance peak is found in both the experiment and the simulation. Figures 19 and 20 indicates the cause of the second resonance peak. The suspended mass is actually oscillating at half of the driving frequency.

The second resonance is a difficult phenomenon to simulate, the fact that the computer simulation does demonstrate the resonance gives credibility to the modeling. Misalignment of the simulation and experimental peaks displays only a slight limitation in the model.

For the second control law an intuitive physical interpretation of an ideal low transmissibility system was investigated. It would be logical to ask the suspension system to keep the suspended mass as close to the design height as possible. To accomplish this the spring would have to have a high spring rate when the mass is moving away from the design height and a low spring rate when it is approaching the design height.

To incorporate the second control law into the hardware the position and slope of the suspended mass signal is required. When the slope and displacement are the same sign the controller must close the orifice. Conversely when the slope and position are of opposite sign the orifice must be opened.

Figure 19 displays computer simulation results for the second control law. The transmissibility plots of the controlled systems are displayed in Figure 20. The systems do show similar characteristics even though the plots do not match. Both models show a significant reduction in transmissibility from the non-controlled systems. Thus, the control law has achieved the goal of reducing the transmissibility.



Figure 17. Control law #1



Figure 18. Transmissibility plot for control law #1



Figure 19. High frequency response for control law #1 experiment



Figure 20. High frequency response for control law #1 simulation



Figure 21. Control law #2



Figure 22. Transmissibility plot for control law #2

CONCLUSIONS

Two main goals were achieved in the active air spring suspension system research presented herein. The first goal was to tune a computational model to an actual system. The tuned model then provides a convenient mechanism to optimize control laws. The other goal was to produce a control law to minimize the transmissibility of the system.

A good match of the computer and hardware models was achieved for the non-controlled cases with some mathematical model manipulation. The controlled cases showed a less accurate fit even though the simulation did demonstrate some of the important trends of the controlled system.

The second control law was effective in lowering the transmissibility of the suspension system. To further minimize transmissibility the control law could be terminated for frequencies above one and a half hertz, (about 9.5 rad/sec). As seen in Figure 20, such a system would provide a very flat transmissibility for input near resonance and give very low transmissibility for high frequencies.

Recommendation for additional research include the following:

1. Extensions of input forms to included non-sinusoidal cases, such that further study is done on the control laws over a broader range of dynamic inputs.

2. Further investigation of the sources of damping is needed to better understand the air spring system behavior.
3. As accumulator volumes grow to large values the contribution to lower natural frequency diminishes. Thus the system does not 'see' the full accumulator. Further study is need to improve modeling for systems with low natural frequency requirements.
4. Further research in the area of handling qualities should be also undertaken.

The computer simulation and the actual quarter vehicle tests demonstrate the potential for a good suspension system without a shock absorber. The fact that transmissibility could be kept low under all driving frequencies shows a great deal of promise for the system

REFERENCES

[1] Claar, Paul W., & Vogel, Jerry M. (1989). A review of active suspension control for on and off-highway vehicles. SAE 982482.

[2] Gillespie, T. D. (1992). Fundamentals of Vehicle Dynamics. Warrendale, PA: Society of Automotive Engineers, Inc.

[3] James, M. L., Smith, G. M., Wolford, J. C., & Whaley, P. W. (1989). Vibration of Mechanical and Structural Systems. New York: Harper & Row.

[4] Sullivan, D. A., & Thompson, P. A. (1984). Real-gas expansion factors for flow nozzles and venturis. Mass Flow Measurements, Vol. 17, pp. 41-48.

[5] Benedict, R. P. (1984). The compressible flow discharge coefficient for a fluid meter. Mass Flow Measurements, Vol. 17, pp. 55-61.

APPENDIX

An area of interest may be the non-linear air spring's effect on the transmissibility plot. If a one degree of freedom system with a mechanical spring and dashpot is taken as the standard the differences can be seen in Figure 21. Note the slope of the curve just before and after resonance.



Figure 23. Transmissibility of air spring vs. mechanical spring

The following is a complete listing of the simulation code. Following the code is a copy of one of the input files used.

```
************************************************************************
        subroutine compd
************************************************************************
        implicit real*8(a-h,l-z)
        integer n
        common/param/var,h,endva,flag,x,endno,n
        common/const/xout,ms,mt,tw,ta,wt,zt,cs,bo,lo,pd,ud,sd,dh,av,dc
        common/const2/temp,gamma,sac,g,z,zdot,tim1,tim2
        common/const3/ap,au,boft,loft,sdft,dhft,poden,tam
        common/const4/xsmax,xsmin,xtmax,xtmin,freq,freq2
        common/const5/slope,yint,po,deno
        common/var/ps,orif,mflow,pa,sh,den,trans,tran1,tran2
        dimension var(14,50)
******    Terrain
        z=ta*sin(tw*x)
        zdot=ta*tw*cos(tw*x)
        zddot=-ta*tw**2*sin(tw*x)
******    System differential equations
        var(8,1)=var(1,2)
        var(8,3)=var(1,4)
         term1=(ps-po)*ap/ms
         term2=-(cs/ms)*(var(1,2)-var(1,4))
         term3=-g
        var(8,2)=term1+term2+term3
         term1=(cs/mt)*(var(1,2)-var(1,4))
         term2=-2.0*wt*zt*(var(1,4)-zdot)
         term3=-wt**2*(var(1,3)-z)
         term4=-(ps-po)*ap/mt
         term5=(ms*g/mt)
        var(8,4)=term1+term2+term3+term4+term5
******    Mass Flow calculations
        sh=dhft+var(1,1)-var(1,3)
        call bagvol(sh,vol)
        den=var(1,5)/(vol)
        ps=(den**gamma)*poden
        pa=(((tam-var(1,5))/av)**gamma)*poden
        pdel=pa-ps
        pdif=dabs(pdel)
        if (abs(1.0-(ps/pa)).lt.0.015) goto 200
        if (pdel.ge.0.0) then
         denup=(tam-var(1,5))/av
         pup=pa-po
         direc=1.0
        else
         denup=den
         pup=ps-po
```

```
         direc=-1.0
        endif
******    Control law goes here (note: diameters are given, not areas)
        orif=bo
******
        beta=orif/ud
        ak=dc/dsqrt(1-(beta**4))
        r=1.0-(pdif/pup)
         term1=(gamma*r**(2.0/gamma))/(gamma-1.0)
         term2=(1.0-beta**4)/(1.0-(beta**4)*(r**(2.0/gamma)))
         term3=(1.0-r**((gamma-1.0)/gamma))/(1.0-r)
        ysq=term1*term2*term3
        y=dsqrt(ysq)
        mflow=ak*y*(orif/2.0/12.0)**2*3.14159*dsqrt(denup*pdif)
        var(8,5)=direc*mflow
        goto 300
200     continue
        var(8,5)=0.0
300     continue
        return
        end

************************************************************************
    subroutine compt
************************************************************************
        implicit real*8(a-h,l-z)
        integer n
        common/param/var,h,endva,flag,x,endno,n
        common/const/xout,ms,mt,tw,ta,wt,zt,cs,bo,lo,pd,ud,sd,dh,av,dc
        common/const2/temp,gamma,sac,g,z,zdot,tim1,tim2
        common/const3/ap,au,boft,loft,sdft,dhft,poden,tam
        common/const4/xsmax,xsmin,xtmax,xtmin,freq,freq2
        common/const5/slope,yint,po,deno
        common/var/ps,orif,mflow,pa,sh,den,trans,tran1,tran2
        dimension var(14,50)
10      format(9(e14.7,2x))
        if(x.eq.0.0) xout2=xout
        xs=var(1,1)
        as=var(8,2)
        xt=var(1,3)
        at=var(8,4)
        mflow=var(8,5)
        if(x.ge.xout2) then
         write(31,10) x,xs*12.0,as,xt*12.0,at,trans,z,mflow,pa/ps
         xout2=xout+x
        endif
*
        if (x.eq.0.0) then
         tim1=0.0
         tim2=0.0
         freq=0.0
```

```
            freq2=0.0
            xsmax=0.0
            xsmin=0.0
            xtmax=0.0
            xtmin=0.0
            trans=0.0
            tran1=0.0
            tran2=0.0
          endif
*
          if ((xs.ge.0.0).and.(var(2,1).le.0.0)) then
            if ((tim1.ne.0.0).and.(x-tim1.gt.0.05)) then
              freq=1.0/(x-tim1)
              write(6,*) 'mass freq =',freq,'     time',x
              tran1=trans
            endif
            tim1=x
          endif
          if ((xt.ge.0.0).and.(var(2,3).le.0.0)) then
            if ((tim2.ne.0.0).and.(x-tim2.gt.0.05)) then
              freq2=1.0/(x-tim2)
c             write(6,*) 'tire freq =',freq2,'     time',x
              tran2=trans
            endif
            tim2=x
          endif
          if (x.gt.7.0) then
            if (xs.gt.xsmax) xsmax=xs
            if (xs.lt.xsmin) xsmin=xs
            if (xt.gt.xtmax) xtmax=xt
            if (xt.lt.xtmin) xtmin=xt
          endif
          axs=((xsmax-xsmin)/2.0)
          axt=((xtmax-xtmin)/2.0)
          if (axt.ne.0.0) then
            trans=(axs)/(axt)
          endif
*
          if(x.ge.endva)then
            flag=-1
            return
          endif
          return
          end

**********************************************************************
      subroutine input
**********************************************************************
          implicit real*8(a-h,l-z)
          integer n
          common/param/var,h,endva,flag,x,endno,n
```

```
      common/const/xout,ms,mt,tw,ta,wt,zt,cs,bo,lo,pd,ud,sd,dh,av,dc
      common/const2/temp,gamma,sac,g,z,zdot,tim1,tim2
      common/const4/xsmax,xsmin,xtmax,xtmin,freq,freq2
      common/const5/slope,yint,po,deno
      dimension var(14,50)
10    format(f10.0)
      read(4,'(i2)') n
      read(4,10) endno
      read(4,10) h
      read(4,10) endva
      read(4,10) x
      read(4,10) xout
      read(4,*)
      read(4,10) ms
      read(4,10) mt
      read(4,*)
      read(4,10) var(1,1)
      read(4,10) var(1,2)
      read(4,10) var(1,3)
      read(4,10) var(1,4)
      read(4,*)
      read(4,10) tw
      read(4,10) ta
      read(4,*)
      read(4,10) wt
      read(4,10) zt
      read(4,10) cs
      read(4,10) slope
      read(4,10) yint
      read(4,*)
      read(4,10) bo
      read(4,10) lo
      read(4,*)
      read(4,10) pd
      read(4,10) ud
      read(4,*)
      read(4,10) sd
      read(4,10) dh
      read(4,10) av
      read(4,10) dc
      read(4,10) temp
      read(4,10) gamma
      read(4,10) sac
      read(4,10) g
      po=2116.2
      deno=.0023769
      return
      end

**********************************************************************
      subroutine main_integration
```

```
***********************************************************************
        implicit real*8(a-h,l-z)
        integer n
        common/param/var,h,endva,flag,x,endno,n
        common/const/xout,ms,mt,tw,ta,wt,zt,cs,bo,lo,pd,ud,sd,dh,av,dc
        common/const2/temp,gamma,sac,g,z,zdot,tim1,tim2
        common/const3/ap,au,boft,loft,sdft,dhft,poden,tam
        common/const4/xsmax,xsmin,xtmax,xtmin,freq,freq2
        common/const5/slope,yint,po,deno
        common/var/ps,orif,mflow,pa,sh,den,trans,tran1,tran2
        dimension var(14,50)
        open (unit=4,file='input.dat',status='unknown')
        call input
        close (4)
*******************************************
        ap=((pd/2.0/12.0)**2)*3.14159
        au=((ud/2.0/12.0)**2)*3.14159
        boft=((bo/2.0/12.0)**2)*3.14159
        loft=((lo/2.0/12.0)**2)*3.14159
        sdft=((sd/2.0/12.0)**2)*3.14159
        dhft=dh/12.0
        poden=po/(deno**gamma)
        ps=(ms*g/ap)+po
        pa=ps
        mflow=0.0
        den=(ps*(1.0/poden))**(1.0/gamma)
        call bagvol(dhft,vol)
        var(1,5)=den*(vol)
        tam=den*(av)+var(1,5)
        write(6,*) 'initial absolute pressure (psi)',pa/(12.0**2)
        write(6,*) 'initial gauge pressure (psi)',(pa-po)/(12.0**2)
        write(6,*) 'desity of air (sulgs/ft^3)',den
*******************************************
        call dode
        return
        end

***********************************************************************
****** Main Program
***********************************************************************
        implicit real*8(a-h,l-z)
        integer n
        common/param/var,h,endva,flag,x,endno,n
        common/const/xout,ms,mt,tw,ta,wt,zt,cs,bo,lo,pd,ud,sd,dh,av,dc
        common/const2/temp,gamma,sac,g,z,zdot,tim1,tim2
        common/const3/ap,au,boft,loft,sdft,dhft,poden,tam
        common/const4/xsmax,xsmin,xtmax,xtmin,freq,freq2
        common/const5/slope,yint,po,deno
        common/var/ps,orif,mflow,pa,sh,den,trans,tran1,tran2
        dimension var(14,50)
        open(31,file='output.dat',status='unknown')
```

```
      call main_integration
      write(6,*) 'Transmisiblity =',(tran1+tran2)/2.0
      close(31)
      stop
      end
************************************************************************
      subroutine bagvol(ht,vol)
************************************************************************
      implicit real*8(a-h,l-z)
      common/const5/slope,yint,po,deno
      if (ht.lt.0.625) then
       vol=(slope*.625)+yint
       goto 10
      endif
      if (ht.lt.1.08333) then
       vol=slope*ht+yint
       goto 10
      endif
      vol=(slope*1.08333)+yint
10    continue
      return
      end
************************************************************************
************************************************************************
******   End of main program, start of dnode
************************************************************************
************************************************************************
      SUBROUTINE PREDI
      implicit real*8(a-h,o-z)
      COMMON/PARAM/VAR,H,ENDVA,FLAG,X,ENDNO,N
      DIMENSION VAR(14,50)
      DO 450 I=1,N
450    VAR(1,I)=(1.5476511d0*VAR(2,I))-(1.8675052d0*VAR(3,I))
     1 +(2.01720690d0*VAR(4,I))-(.69735280d0*VAR(5,I))+
     2 H*((2.00224730d0*VAR(9,I))-(2.03168770d0*VAR(10,I))
     3 +(1.81861080d0*VAR(11,I))-(0.714320050d0*VAR(12,I)))
      RETURN
      END
************************************************************************
      SUBROUTINE CORRT
      implicit real*8(a-h,o-z)
      COMMON/PARAM/VAR,H,ENDVA,FLAG,X,ENDNO,N
      DIMENSION VAR(14,50)
      DO 462 I=1,N
      VAR(1,I)=VAR(2,I)+H*((.3750d0*VAR(8,I))+(.791666670d0*VAR(9,I))
     1 -(.20833330d0*VAR(10,I))+(.0416666670d0*VAR(11,I)))
462    CONTINUE
      RETURN
      END
************************************************************************
      SUBROUTINE INITA
```

```
      implicit real*8(a-h,o-z)
      COMMON/PARAM/VAR,H,ENDVA,FLAG,X,ENDNO,N
      DIMENSION VAR(14,50),A(4),B(4),C(4)
      A(1)=.50d0
      A(2)=.292893220d0
      A(3)=1.70710680d0
      A(4)=.166666670d0
      B(1)=1.0d0
      B(2)=.292893220d0
      B(3)=1.70710680d0
      B(4)=.33333330d0
      C(1)=.50d0
      C(2)=.292893220d0
      C(3)=1.70710680d0
      C(4)=.50d0
      DO 402 I=1,N
      VAR(6,I)=0.0
402   CONTINUE
      J=4
      GO TO 410
403   DO 407 K=1,4
      DO 404 I=1,N
      CK=H*VAR(8,I)
      R=A(K)*CK-B(K)*VAR(6,I)
      VAR(1,I)=VAR(1,I)+R
404   VAR(6,I)=VAR(6,I)+3.0d0*R-C(K)*CK
      IF(K-1)405,405,413
413   IF(K-3)406,405,406
C NEW VALUE OF X
405   X=X+H/2.0d0
      CALL COMPD
      GO TO 407
406   CALL COMPD
407   CONTINUE
410   DO 408 I=1,N
      VAR(J+1,I)=VAR(1,I)
      VAR(J+8,I)=VAR(8,I)
408   CONTINUE
C ADD THIS CALL TO THE OUTPUT ROUTINE
      CALL COMPT
      J=J-1
      IF(J.GT.0) GO TO 403
      RETURN
      END
************************************************************************
      SUBROUTINE DODE
      implicit real*8(a-h,o-z)
      COMMON/PARAM/VAR,H,ENDVA,FLAG,X,ENDNO,N
      DIMENSION VAR(14,50)
C INITIALIZE
      FLAG=0.
```

```
      IF(ENDNO.GT.0.0) H=(ENDVA-X)/ENDNO
C PREPARE FOR RKG
      CALL COMPD
c the first call to compt occurs in INITA
c       CALL COMPT
      CALL INITA
510   X=X+H
      CALL PREDI
      CALL COMPD
      CALL CORRT
      CALL COMPD
      CALL COMPT
      IF(FLAG.LT.0) RETURN
528   J=13
523   DO 524 I=1,N
524   VAR(J+1,I)=VAR(J,I)
      J=J-1
      IF(J) 510,510,523
      END
*************************************************************************
```

```
5          n - number of equations for dnode
0.0        endno - # of steps (0.0 for using h and enval)
0.0002     h - Step size
12.0       endva - time value that dnode stops
0.0        x - starting time for dnode
0.01       xout - output time step
**********
4.6        ms - sprung mass (slugs)
1.0        mt - tire mass (slugs)
**********
0.0        xs - var(1,1) - initail sprung mass displacement (ft)
0.0        vs - var(1,2) - initial sprung mass velocity (ft/sec)
0.0        xt - var(1,3) - initial tire mass displacement (ft)
0.0        vt - var(1,4) - initial tire mass velocity (ft/sec)
**********
4.0        tw - terrain natural frequency (rad/sec)
0.0416667       ta - terrain amplitude (ft)   0.0416667
**********
70.0       wt - tire's natural frequency (rad/sec)
0.15       zt - tire's damping ratio
11.0       cs - air spring's damping coefficient (lb sec/ft) (9-16)
0.06       slope - volume function slope
-0.015     yint - volume function intersept
**********
0.0        bo - big orifice diameter (INCHES)
0.0        lo - little orifice diameter (INCHES)
**********
2.35       pd - piston diameter (INCHES)
1.0        ud - upstream diameter (INCHES)
**********
3.9        sd - spring diameter (INCHES)
10.5       dh - design height (INCHES)
0.08       av - accumulator volume (ft^3)  .055
0.6        dc - discharge coefficent
520.0      temp - temperature (deg R)
1.4        gamma - specific heat ratio
1716.0     sac - specific air constant
32.17      g - acceleration of gravity (ft/sec^2)
```